Micro-code for Triangle Scan Driver,  8/11/08 
Microprocessor code listing (air-cooled version).  

      ;THIS PROGRAM IS 8051/8031 CODE FOR THE DIGITAL TRIANGLE GENERATOR

      ;AND COIL DRIVER UNIT FOR USE AT UNIVERSITY OF CHICAGO.

      ;THIS CODE RUNS ON CIRCUIT CARD DU P/N 17110

      ;BUILT IN THE UNIVERSITY OF DENVER CHEMISTRY AND ENGINEERING

      ;DEPARTMENTS FOR USE IN EPR RAPID SCAN SPECTROSCOPY.

      ;

      ;WRITTEN BY R. W. QUINE

      ;

      ;

             .ORG 0

             ;EXTERNAL RAM LOCATION ASSIGNMENTS

       ;none

      ;RESET VECTOR

             .ORG 0

        JMP START       ;RESET VECTOR

      ;

             .ORG 3

        JMP INTSER      ;INTERRUPT SERVICE VECTOR FOR IE0

             .ORG 13H

      ;  JMP INT1X        ;INTERRUPT SERVICE VECTOR FOR IE1 (NOT USED)

      ;

      ;ON-CHIP CACHE MEMORY

      ;BIT ADDRESSABLE

        .ORG 20H

        FLAGS: .BLKB 1  MISC. FLAGS

        ;NONE USED

       SPHC: .BLKB 2    ;PHASE CONSTANT STORAGE

       ONE80: .BLKB 1   ;BIT 0 = THE 180 DEGREE BIT

       CMODEL: .BLKB 1  ;MODEL OF STEP COUNTER IN HARDWARE

        ;

        ;30H TO 4FH RESERVED FOR STACK

        ;50H TO 7FH AVAILABLE FOR CACHE STORAGE

              .ORG 50H

       STATUS: .BLKB 1  ;INTERNAL STATUS

       ;BIT 0 = 

       ;BIT 1 = 

       ;BIT 2 = 

       ;BIT 3 = 

       ;BIT 4 = 

       ;BIT 5 = WAITING FOR BRUKER = 1

       ;BIT 6 = RECEIVED BLOCK OF 6 BYTES AVAILABLE

       ;BIT 7 = RECEIVED CHARACTER AVAILABLE

       ISAVE5: .BLKB 5  ;TEMP SAVE FOR INTERRUPTS

       ISAVE:  .BLKB 5  ;TEMP SAVE FOR INTERRUPTS

       IMAGE:  .BLKB 1    ;IMAGE OF U33 OUTPUT REGISTER

       BLOCK6: .BLKB 6    ;BLOCK OF 6 BYTES RAW FROM BRUKER

       ;REARRANGED RAW BRUKER DATA INTO LO-HI FORMAT

       PH: .BLKB 2

       SF: .BLKB 2

       SW: .BLKB 2

       ;INPUTS AND OUTPUTS FOR MULTIPLY SUBROUTINE

       MULIN2: .BLKB 2   ;2 BYTE INPUT FOR 12 BIT DATA

       MULIN4: .BLKB 4   ;4 BYTE INPUT FOR THE MULTIPLYING CONSTANT

       ACCUM4: .BLKB 4   ;4 BYTE MULTIPLY ANSWER

       ;SWEEP AMPLITUDE (SWEEP WIDTH) NUMBERS

       ;LIMIT CONSTANTS

       A1LIML:   .SET FFH       ;COIL 1 AMPLITUDE LIMIT LO

       A1LIMH:   .SET 0FH       ;COIL 1 AMPLITUDE LIMIT HI

                                ;COIL 1 CONSTANT IS 10.8 G/A

                                ;COIL 1 IS FOR THE 16 MM RESONATOR

       A2LIML:   .SET FFH       ;COIL 2 AMPLITUDE LIMIT LO

       A2LIMH:   .SET 0FH       ;COIL 2 AMPLITUDE LIMIT HI

                                ;COIL 2 NOT YET PROGRAMED.

       ;LIMITS ARE COMPUTED AS CC * FSApp /.005 

       ;WHERE CC IS THE COIL CONSTANT IN G/A

       ;FSApp IS THE MAX AMP PP OF THE TRIANGLE DRIVER, NOW 2.0 App

       ;.005 IS THE BRUKER LSB IN GAUSS

       ;IF THIS COMPUTES TO GREATER THAN 4095 THEN USE 4095 (0FFFH)

       A1LIM: .BLKB 2    ;COIL 1 LIMIT STORAGE

       A2LIM: .BLKB 2    ;COIL 2 LIMIT STORAGE

       ;SWEEP WIDTH D/A SCALING CONSTANTS

       SSC1L:    .SET AAH       ;SWEEP SCALING CONSTANT FOR COIL 1 LO

       SSC1H:    .SET 3CH       ;SWEEP SCALING CONSTANT FOR COIL 1 HI

                                ;CC FOR COIL 1 IS 10.8 G/A

       SSC2L:    .SET AAH       ;SWEEP SCALING CONSTANT FOR COIL 2 LO

       SSC2H:    .SET 3CH       ;SWEEP SCALING CONSTANT FOR COIL 2 HI

                                ;COIL 2 SET TO SAME AS COIL 1 FOR NOW.

       ;SWEEP SCALING CONSTANTS COMPUTED AS FOLLOWS:

       ;SSC =  4095 * 2^14 * .005 / (CC * FSApp)

       ;WHERE 4095 IS D/A FULL SCALE 12 BIT NUMBER

       ;2^14 IS CHOSEN TO FILL UP THE 32 bit MULTIPLY ROUTINE WITH 

       ;THE MOST DIGITS OF RESOLUTION AND NOT OVERFLOW AND NOT REQUIRE

       ;MORE THAN 16 BITS OF SCALING CONSTANT FOR CC AT LEAST 7 G/A

       ;.005 IS THE BRUKER LSB IN GAUSS

       ;CC IS THE COIL CONSTANT 

       ;FSApp IS THE FULL SCALE AMPS OF THE TRIANGLE DRIVER, NOW 2.0 App

       SSC1: .BLKB 2    ;COIL 1 SCALING CONSTANT STORAGE

       SSC2: .BLKB 2    ;COIL 2 SCALING CONSTANT STORAGE

       ;SWEEP FREQUENCY MULTIPLY CONSTANT

       ;DECIMAL VALUE IS 43808.7, HEX VALUE IS AB21

       SWFMCL:   .SET 21H

       SWFMCM:   .SET ABH

       SWFMCH:   .SET 00H

       ;THIS NUMBER IS 1*510*2^32 / 50,000,000.

       ;THE PARTS OF THIS ARE:

       ;1 = BRUKER LSB RESOLUTION IN HZ

       ;510 = STEP CLOCK TO SWEEP FULL PERIOD RATIO. 

       ;THE TRIANGLE IS GENERATED BY 510 STEPS.

       ;2^32 = FRACTIONAL RESOLUTION OF AD9830 FREQ. SYNTHESIZER

       ;50,000,000 = 50 MHZ TIME BASE OF AD9830.

       ;SPECJET PHASE MULTIPLY CONSTANT

       ;DECIMAL VALUE IS 9287, HEX VALUE IS 2447H

       SPHCL:    .SET 47H

       SPHCH:    .SET 24H

       ;THIS IS 0.1 * 2^16 * 510/359.9 

       ;WHERE 0.1 IS THE BRUKER 1 LSB RESOLUTION IN DEGREES

       ;2^16 SCALES UP THE MULTIPLY RESOLUTION TO 32 BITS

       ;510 IS THE NUMBER OF SYNTHESIS (COUNTER) STEPS PER CYCLE

       ;359.9 IS THE FULL SCALE DEGREES PER CYCLE

       ;NOW THE PHASE CALIBRATION CONSTANT

       ;THIS SETS THE OFFSET SO THAT A PHASE OF 0 DEGREES

       ;FROM BRUKER PUTS THE SPECJET TRIGGER EXACTLY ON

       ;THE LOWER APOGEE OF THE CURRENT WAVEFORM

       ;THE FIXED INDEX POINT IS A COUNTER VALUE OF 128 WITH

       ;THE 180 BIT =1. THE OFFSET CAN BE SET FROM 0 TO 255

       ;SO THAT A NOMINAL OFFSET OF 127 WOULD PLACE THE

       ;TRIGGER AT THE LOWER APOGEE

       PHOFFS:  .SET 127        ;PHASE OFFSET CALIBRATION

                                ;SET 127  TO ACHIEVE

                                ;TRIGGER AT LOWER APOGEE OF

                                ;THE CURRENT WAVEFORM WHEN

                                ;BRUKER PHASE = 0

       ;CACHE USAGE SHOULD BE LESS THAN 7FH HERE

       ;

       ;VALUE EQUIVALENTS SET HERE FOR STANDARD 8051 LOCATIONS

       SP:       .SET 81H

       PSW:      .SET 0D0H

       P1:       .SET 90H

       P2:       .SET 0A0H

       IE:       .SET 0A8H

       IP:       .SET 0B8H

       ACC:      .SET 0E0H

       DPTRL:    .SET 82H

       DPTRH:    .SET 83H

       B:        .SET F0H

       TCON:     .SET 88H

       TMOD:     .SET 89H

       TL1:      .SET 8BH

       TH1:      .SET 8DH

       ;CHIP SELECTS - VALUES SET ON P1 LOWER 5 BITS

       ;UPPER 3 BITS MAY BE USED ELSEWHERE

       CSLSB:   .SET 0

       CSMSB:   .SET 1

       CSOUT:   .SET 2

       CSPH:    .SET 3

       CSWRN:   .SET 4

       CSLOD:   .SET 5

       CSLAT:   .SET 6

       CSCOM:   .SET 7

       CSOFF:   .SET 16

       ;COM PORT REGISTERS IN SCN2681 UART CHIP

       MR1A:    .SET 0

       SRA:     .SET 1

       ISR:     .SET 5

       MR1B:    .SET 8

       SRB:     .SET 9

       CSRA:    .SET 1

       CRA:     .SET 2

       THRA:    .SET 3

       ACR:     .SET 4

       IMR:     .SET 5

       CSRB:    .SET 9

       CRB:     .SET 10

       THRB:    .SET 11

       OPCR:    .SET 13

       RHRA:    .SET 3

       RHRB:    .SET 11

;

               .ORG 100H

      ;

      ;MAIN PROGRAM STARTS HERE

      ;

START:  CLR A                   ;GET ZERO

        MOV PSW,A               ;RESET FLAGS IN PROCESSOR STATUS WORD

        MOV R0,#<STATUS         ;POINT TO STATUS

        MOV @R0,A               ;ZERO OUT STATUS

        MOV SP,#2FH             ;INITIALIZE STACK POINTER- 30H IS FIRST

                                ;LOCATION USED

        MOV A,#01101000B        ;SET LIGHTS AND CONTROL OUTPUTS

ERRENT: CALL OUT                ;GO LOAD U33 AND IMAGE

        CALL IN9830             ;GO INITIALIZE THE FREQUENCY SYNTHESIZER

        CALL INTDA              ;GO INITIALIZE THE AMPLITUDE D/A TO ZERO

        CALL INTCOM             ;INITIALIZE COM CHIP

        CALL STORE              ;GO STORE VARIOUS CONSTANTS IN RAM

        MOV IE,#10000001B       ;EN INT. FOR COM ONLY

        ;MOV IE,#10000101B      ;EN INT. FOR COM AND INT1 (NOT USED)

        MOV IP,#00000101B       ;SET INTERRUPT PRIORITY

        MOV TMOD,#10010000B     ;SET TIMER MODE

        MOV TCON,#01000100B     ;SET TIMER CONTROL

        MOV R0,#<STATUS         ;SET POINTER

        MOV A,@R0               ;GET STATUS

        ORL A,#00100000B        ;OR IN WAITING FOR BRUKER BIT

MLOOPB: MOV @R0,A               ;STORE IT

        CALL SETWFB             ;GO SET OR RESET THE WAITING FOR BRUKER LIGHT

        MOV R0,#<BLOCK6         ;SET POINTER TO BLOCK STORAGE

MLOOPE: MOV R1,#<STATUS         ;SET POINTER

MLOOP:  MOV A,@R1               ;GET STATUS

        JB ACC.6,BLKFUL         ;JUMP IF BLOCK OF 6 IS FULL

        JB ACC.7,GETBYT         ;JUMP IF BYTE RECIEVED FROM BRUKER

        JMP MLOOP               ;WAIT FOR BYTE

SETWFB: ;HERE TO SET OR RESET THE WAITING FOR BRUKER LIGHT

        ;ACCORDING TO BIT 5 OF STATUS

        MOV R0,#<STATUS         ;POINT TO STATUS

        MOV A,@R0               ;GET STATUS

        JB ACC.5,WFB            ;JUMP IF WAITING FOR BRUKER

        MOV R0,#<IMAGE          ;POINT TO IMAGE OF U33

        MOV A,@R0               ;GET IMAGE

        ORL A,#1                ;SET LSB FOR GREEN LIGHT- NOT WAITING

        CALL OUT                ;GO LOAD U33 AND UPDATE IMAGE

        RET                     ;RETURN FROM SETWFB

WFB:    MOV R0,#<IMAGE          ;POINT TO IMAGE OF U33

        MOV A,@R0               ;GET IMAGE

        ANL A,#11111110B        ;SET LSB FOR RED LIGHT- WAITING

        CALL OUT                ;GO LOAD U33 AND UPDATE IMAGE

        RET                     ;RETURN FROM SETWFB

GETBYT:        ;HERE TO GET A BYTE FROM BRUKER AND PUT IT IN BLOCK

        ANL A,#01111111B        ;KILL DA BIT

        MOV @R1,A               ;STORE NEW STATUS

        ORL PSW,#00001000B      ;SELECT INTERRUPT REGISTERS

        MOV A,R3                ;GET RECEIVED BYTE

        ANL PSW,#11110111B      ;BACK TO FOREGROUND

        MOV @R0,A               ;STORE NEW BYTE IN BLOCK

        INC R0                  ;SET POINTER FOR NEXT ONE

        MOV A,R0                ;GET POINTER VALUE

        XRL A,#<BLOCK6        ;TEST FOR BLOCK FULL

        JZ  FULLBK              ;JUMP IF FULL

        JMP MLOOPE              ;JUMP TO MAIN LOOP 

FULLBK: MOV A,@R1               ;GET STATUS

        ORL A,#01000000B        ;SET BLOCK FULL BIT

        MOV @R1,A               ;STORE NEW STATUS

        JMP MLOOP               ;BACK TO MAIN LOOP

BLKFUL:         ;HERE IF BLOCK OF 6 RAW BYTES FROM BRUKER HAS BEEN RECD

        CALL PARSE              ;PARSE OUT RAW BRUKER BYTES TO MEMORY

        CALL SETSW              ;GO SET HARDWARE FOR SWEEP WIDTH

        CALL SETSF              ;GO SET HARDWARE FOR SWEEP FREQUENCY

        CALL SETPH              ;GO SET HARDWARE FOR SPECJET PHASE

        MOV R0,#<STATUS         ;SET POINTER TO STATUS

        MOV A,@R0               ;GET STATUS

        ANL A,#10011111B        ;TURN OFF WAITING FOR BRUKER BIT

                                ;AND THE BLOCK FULL BIT

        JMP MLOOPB              ;REENTER MAIN LOOP

PARSE:  MOV R2,#6               ;SET LOOP COUNTER

        MOV R0,#<BLOCK6       ;SET POINTER TO BLOCK STORAGE

        MOV R1,#<PH             ;POINT TO LO-HI FORMAT STORAGE

PARL1:  MOV A,@R0               ;GET BRUKER RAW

        MOV @R1,A               ;STORE IN LO-HI FORMAT LOCATION

        DEC R0                  ;NEXT IN BLOCK STORAGE

        INC R1                  ;NEXT LO-HI FORMAT

        DJNZ R2,PARL1           ;TEST FOR END OF LOOP

        RET                     ;RETURN FROM PARSE

SETSF:  ;HERE TO COMPUTE THE SWEEP FREQUENCY SETTING TO BE APPLIED

        ;TO THE AD9830 FREQUENCY SYNTHESIZER

        ;BRUKER OR PC DATA IS THE OFFSET TO BE ADDED TO 500 HZ. THE FINAL

        ;FREQUENCY RANGE IS 500 TO 4595 HZ IN ONE HZ STEPS.

        MOV R0,#<SF             ;POINT TO BRUKER SWEEP FREQUENCY DATA LOW

        MOV R1,#<MULIN2       ;POINT TO MULIN2 HI

LOWOK1: INC R0                  ;NOW TO TEST UPPER BRUKER BYTE

        MOV A,@R0               ;GET IT

LOWOK:  ANL A,#11110000B        ;UPPER BITS, 4096 AND GREATER

        JNZ SFRED               ;JUMP IF FREQ >= 4596 HZ TO RED LIGHT

        MOV A,<IMAGE            ;GET U33 IMAGE

        ORL A,#00001000B        ;SET FOR GREEN FREQUENCY LIGHT

        CALL OUT                ;LOAD U33 TO SET GREEN LIGHT

        MOV A,@R0               ;GET FREQ HI BYTE

        MOV @R1,A               ;PUT IT IN MULIN2 HI

        DEC R0                  ;POINT TO FREQUENCY LOW

        DEC R1                  ;POINT TO MULIN2 LO

        MOV A,@R0               ;GET LO

        MOV @R1,A               ;STORE IT IN MULIN2 LO

        MOV R0,#<MULIN4         ;POINT TO MULIN4 LO

        MOV @R0,#SWFMCL         ;LOAD LO CONSTANT

        INC R0                  ;NOW MID LO

        MOV @R0,#SWFMCM         ;LOAD MID LO

        INC R0                  ;NOW MID HI

        MOV @R0,#SWFMCH         ;LOAD MID HI

        INC R0                  ;NOW HI

        MOV @R0,#0              ;LOAD HI

        CALL MULTPY             ;GO DO THE MULTIPLY

        JC ERRO5                ;JUMP TO ERROR TRAP IF OVERFLOW

        ;NOW ADD 500 HERTZ

        ;500 HZ IN SYNTHESIZER = 014E3BCD

        MOV R0,#<ACCUM4         ;POINT TO PREVIOUS MULTIPLY RESULT LOW

        CLR C                   ;CLEAR CARRY

        MOV A,#CDH              ;LOW BYTE 

        ADDC A,@R0              ;ADD LOW BYTE

        MOV @R0,A               ;STORE LOW

        INC R0                  ;NEXT BYTE

        MOV A,#3BH              ;NEXT BYTE

        ADDC A,@R0              ;NEXT ADD

        MOV @R0,A               ;STORE

        INC R0                  ;NEXT

        MOV A,#4EH              ;NEXT

        ADDC A,@R0              ;NEXT ADD

        MOV @R0,A               ;STORE

        INC R0                  ;NEXT

        MOV A,#01H              ;NEXT

        ADDC A,@R0              ;LAST ADD

        MOV @R0,A               ;STORE LAST BYTE

        ;NOW LOAD THE FREQUENCY SYNTHESIZER

        MOV R1,#0               ;SET ADDRESS STARTING VALUE

        MOV R0,#<ACCUM4         ;POINT TO DATA LO

        CALL LOAD4              ;GO LOAD 4 BYTES INTO 9830 CARD

        RET                     ;RETURN FROM SETSF

SFRED:  MOV A,<IMAGE            ;GET U33 IMAGE

        ANL A,#11110111B        ;SET FOR RED

        CALL OUT                ;LOAD U33

        RET                     ;RETURN FROM SETSF WITHOUT SETTING

                                ;HARDWARE

ERRO5:  JMP ERROR5              ;ERROR TRAP

LOAD4:  MOV R2,#2               ;SET LOOP COUNTER

LOAD4L: MOV A,#CSLAT            ;CHIP SELECT FOR LATCH

        CALL SETCS              ;SET CHIP SELECT

        MOV A,@R0               ;GET DATA

        MOVX @R0,A              ;LOAD LATCH

        movx @r0,a

        INC R0                  ;NOW MID LO

        MOV A,#CSLOD            ;CHIP SELECT FOR LOAD

        CALL SETCS              ;SET CHIP SELECT

        MOV A,@R0               ;GET DATA

        MOVX @R0,A              ;LOAD LOAD

        MOVX @R0,A              ;AGAIN AFTER BUS HAS SETTLED

        MOV A,#CSWRN            ;CHIP SELECT FOR WRN

        CALL SETCS              ;SET CHIP SELECT

        MOV A,R1                ;GET ADDRESS                

        MOVX @R0,A              ;WRITE 2 BYTES INTO 9830

        MOVX @R0,A              ;AGAIN

        INC R0                  ;POINT TO NEXT SOURCE BYTE

        INC R1                  ;POINT TO NEXT DESTINATION ADDRESS

        DJNZ R2,LOAD4L          ;UNTIL BOTH ARE DONE

        MOV A,#CSOFF            ;CS ALL OFF

        CALL SETCS              ;GO SET CS

        RET                     ;RETURN FROM LOAD4

IN9830: MOV R0,#<IMAGE          ;POINT TO IMAGE OF U33

        MOV A,@R0               ;GET IMAGE

        ANL A,#11011111B        ;BIT TO RESET 9830

        CALL OUT                ;LOAD U33 AND IMAGE

        MOV A,@R0               ;AGAIN GET IMAGE

        ORL A,#00100000B        ;NEUTRALIZE RESET BIT

        CALL OUT                ;LOAD U33 AND IMAGE

        CALL ZACC               ;ZERO OUT ACCUM4

        MOV R1,#0               ;START WITH FREQ 0 REGISTERS

        MOV R0,#<ACCUM4         ;POINT TO DATA LO

        CALL LOAD4              ;GO LOAD 4 BYTES INTO 9830 CARD

        MOV R0,#<ACCUM4         ;POINT TO DATA LO

        CALL LOAD4              ;4 MORE BYTES FOR FREQ 1 REGISTERS

        MOV R0,#<ACCUM4         ;POINT TO DATA LO

        CALL LOAD4              ;GO LOAD 4 PHASE BYTES INTO 9830 CARD

        MOV R0,#<ACCUM4         ;POINT TO DATA LO

        CALL LOAD4              ;4 MORE PHASE BYTES

        RET                     ;RETURN FROM IN9830

SETSW:  ;HERE TO COMPUTE AMPLITUDE CONTROL VALUE AND LOAD IT

        ;INTO THE D/A CONTROLLER 

GETP1:  MOV A,P1                ;GET P1 PORT

        JNB ACC.6,COIL1         ;JUMP TO COIL 1

        JNB ACC.7,COIL2         ;JUMP TO COIL 2

        JMP GETP1               ;SWITCH BOUNCING, TRY AGAIN

COIL2:  MOV R0,#<A2LIM          ;GET LIMIT LO ADDRESS

        MOV R1,#<SSC2           ;GET SCALING CONSTANT LO ADDRESS

        JMP LMD                 ;GO CHECK LIMITS AND SETUP FOR MULTIPLY

                                ;AND DO MULTIPLY AND LOAD D/A

COIL1:  MOV R0,#<A1LIM          ;GET LIMIT LO

        MOV R1,#<SSC1           ;GET SCALING CONSTANT LO ADDRESS

LMD:    CLR C                   ;CLEAR BORROW

        MOV A,@R0               ;GET LIMIT LO

        SUBB A,<SW              ;SUBTRACT BRUKER LO

        INC R0                  ;NOW HI

        MOV A,@R0               ;GET HI

        SUBB A,<SW            ;SUBTRACT HI

        JC REDLGT               ;IF OUT OF LIMIT JUMP TO RED LIGHT

        MOV A,<IMAGE            ;GET U33 IMAGE

        ORL A,#00000100B        ;SET FOR GREEN AMPLITUDE LIGHT

        CALL OUT                ;GO LOAD U33

        CALL LDIN4              ;LOAD MULIN4 WITH @R1 SCALING

        MOV R1,#<SW             ;POINT TO RAW BRUKER DATA

        CALL LDIN2              ;GO LOAD INTO MULIN2

        CALL MULTPY             ;GO MULTIPLY

        MOV R0,#<ACCUM4         ;SET POINTER

        CALL SHIFT4             ;SHIFT ACCUM4 ONE BIT LEFT (X2)

        MOV R0,#<ACCUM4         ;RESTORE POINTER

        CALL SHIFT4             ;AGAIN (TOTAL OF X4)

        MOV A,<ACCUM4         ;GET HI BYTE

        ANL A,#00001111B        ;SELECT LOWER NIBBLE

        MOV R3,A                ;LOAD UPPER NIBBLE OF HOLDING REGISTER

        MOV A,<ACCUM4         ;GET MID HI BYTE

        MOV R2,A                ;INTO HOLDING REGISTER

        CALL LOADDA             ;GO LOAD THE D/A

        RET                     ;RETURN FROM SETSW

LDIN4:  MOV R0,#<MULIN4         ;POINT TO MULTIPLY 4 BYTE INPUT

        MOV R2,#2               ;SET LOOP COUNTER

LMDLOP: MOV A,@R1               ;GET SCALE CONSTANT

        MOV @R0,A               ;PUT IT IN MULIN4

        INC R0                  ;NEXT UP

        INC R1                  ;NEXT

        DJNZ R2,LMDLOP          ;UNTIL DONE

        CLR A                   ;GET ZERO

        MOV @R0,A               ;LOAD MULIN4 MID HI

        INC R0                  ;NOW HI

        MOV @R0,A               ;LOAD HI BYTE

        RET                     ;RETURN FROM LDIN4

LDIN2:  MOV R0,#<MULIN2         ;POINT TO MULTIPLY 2 BYTE INPUT

        MOV R2,#2               ;SET LOOP COUNTER

LMDL2:  MOV A,@R1               ;GET DATA

        MOV @R0,A               ;PUT IT IN MULIN2

        INC R0                  ;NEXT UP

        INC R1                  ;NEXT

        DJNZ R2,LMDL2           ;UNTIL DONE

        RET                     ;RETURN FROM LDIN2

REDLGT: MOV A,<IMAGE            ;GET U33 IMAGE

        ANL A,#11111011B        ;SET FOR RED AMPLITUDE OUT OF RANGE

        CALL OUT                ;GO SET THE LIGHT

        RET                     ;RETURN FROM SETSW

PHLITE: MOV A,<IMAGE            ;GET U33 IMAGE

        ANL A,#11111101B        ;SET FOR RED PHASE OUT OF RANGE

        CALL OUT                ;GO SET THE LIGHT

        RET                     ;RETURN FROM SETPH

INTDA:  MOV R2,#0               ;SET HOLDING REGISTER TO ZERO

        MOV R3,#0               ;UPPER HOLDING REGISTER = ZERO

        CALL LOADDA             ;GO LOAD THE D/A

        RET                     ;RETURN FROM INTDA

LOADDA: ;HERE TO LOAD THE D/A U24 WITH DATA FROM R2 AND R3

        MOV A,#CSLSB            ;CHIP SELECT FOR D/A LOW BYTE

        CALL SETCS              ;SET CHIP SELECT

        MOV A,R2                ;GET DATA LO

        MOVX @R1,A              ;LOAD D/A LOW, R1 IS DUMMY

        MOV A,#CSMSB            ;CHIP SELECT FOR D/A HI BYTE

        CALL SETCS              ;SET CHIP SELECT

        MOV A,R3                ;GET DATA HI

        MOVX @R1,A              ;LOAD D/A HI

        MOV A,<IMAGE            ;GET IMAGE OF U33

        ANL A,#10111111B        ;LDAC TO LOW

        CALL OUT                ;LOAD U33 AND IMAGE

        MOV A,<IMAGE            ;GET IMAGE

        ORL A,#01000000B        ;LDAC BACK UP TO NEUTRAL

        CALL OUT                ;LOAD U33 AND IMAGE

        RET                     ;RETURN FROM LOADDA

SETPH:  ;HERE TO COMPUTE AND LOAD SPECJET TRIGGER PHASE INTO HARDWARE

        ;ALSO TEST FOR BRUKER DATA > 359.9 DEGREES

        CLR C                   ;CLEAR BORROW

        MOV R1,#<PH             ;POINT TO BRUKER PHASE DATA LOW

        MOV A,#0FH              ;LOW BYTE OF 3599 = E0F

        SUBB A,@R1              ;SUBTRACT BRUKER PHASE

        INC R1                  ;NOW HI BYTE

        MOV A,#0EH              ;UPPER BYTE OF E0F

        SUBB A,@R1              ;SUBTRACT UPPER

        JC PHLITE               ;JUMP TO TURN ON RED OUT OF RANGE LIGHT

        ;HERE IF PHASE IN RANGE 0 TO 359.9 DEGREES

        MOV A,<IMAGE            ;GET U33 IMAGE

        ORL A,#00000010B        ;SET FOR GREEN PHASE LIGHT

        CALL OUT                ;GO LOAD U33

        MOV R1,#<SPHC           ;SET POINTER TO PHASE CONSTANT

        CALL LDIN4              ;GO LOAD IT INTO MULIN4

        MOV R1,#<PH             ;POINT TO BRUKER PHASE DATA

        CALL LDIN2              ;LOAD IT INTO MULIN2

        CALL MULTPY             ;GO MULTIPLY

        CALL PHINDX             ;GO SET THE PHASE CALIBRATION INDEX

        MOV A,<ACCUM4         ;GET UPPER BYTE

        JZ NOUPER               ;JUMP IF NO UPPER BIT

        ANL A,#11111110B        ;SELECT BITS ABOVE 256

        JNZ PHERR               ;JUMP IF ERROR

        MOV R2,#255             ;SET FOR MAX

        CALL MODELC             ;GO INCREMENT THE MODEL COUNTER

        MOV R2,#1               ;ONE MORE FOR 256

        CALL MODELC             ;INCREMENT

NOUPER: MOV R2,<ACCUM4        ;GET LOWER BYTE

        CALL MODELC             ;GO INCREMENT THE REST OF IT

        ;NOW TO LOAD THE HARDWARE WITH THE RESULTS

        MOV A,#CSPH             ;GET CHIP SELECT FOR PHASE WORD

        CALL SETCS              ;SET THE CS FOR U17

        MOV A,<CMODEL           ;GET THE MODEL COUNTER VALUE

        MOVX @R1,A              ;LOAD U17 - R1 IS DUMMY

        MOV A,<ONE80            ;GET THE 180 DEGREE BIT

        SWAP A                  ;PUT THE BIT IN THE HARDWARE POSITION

        ANL A,#00010000B        ;JUST THE 180 BIT

        MOV R5,A                ;TEMP STORE

        MOV A,<IMAGE            ;GET THE U33 IMAGE

        ANL A,#11101111B        ;KILL THE OLD 180 BIT

        ORL A,R5                ;OR IN THE NEW 180 BIT

        CALL OUT                ;GO LOAD U33

        RET                     ;RETURN FROM SETPH

PHERR:  MOV A,#10               ;DIAGNOSTIC ERROR CODE

        JMP ERROR               ;ERROR TRAP

PHINDX: ;THIS SETS THE PHASE TO AN INITIAL VALUE OF NOMINALLY

        ;ZERO OFFSET BY A CALIBRATION VALUE. IF THE CAL VALUE

        ;SET IN PHOFFS IS 127 THE PHASE IS NOMINALLY ZERO.

        ;ADJUST PHOFFS  FROM 127 TO MAKE THE TRIGGER

        ;EXACTLY ZERO (LOWER APOGEE OF CURRENT TRIANGLE)

        ;WHEN BRUKER PHASE SETTING IS ZERO.

        MOV <CMODEL,#127        ;INITIAL INDEX POINT

        MOV <ONE80,#1           ;SET 180 BIT

        MOV R2,#PHOFFS          ;GET THE PHASE OFFSET

        CALL MODELC             ;GO INCREMENT THE MODEL COUNTER

        RET                     ;RETURN FROM PHINDX

MODELC: ;THIS IS A MODEL OF THE HARDWARE COUNTER

        ;THE MODEL COUNTER IS AT RAM LOCATION CMODEL

        ;IT COUNTS TO 255 THEN 1 TO  255

        ;IT NEVER HAS ZERO STATE

        ;THE TRANSITION FROM 255 TO 1 CAUSES THE 180 BIT TO TOGGLE

        ;THE 180 BIT IS ONE80.0

        ;A CALL TO THIS ROUTINE INCREMENTS THE MODEL BY THE VALUE

        ;IN R2

        MOV A,R2                ;GET INCREMENT COUNTER

        JZ RETU                 ;RETURN IF ZERO - NO INCREMENT REQUIRED

        MOV A,<CMODEL           ;GET THE COUNTER MODEL

MODLP:  INC A                   ;INCREMENT 

        JZ FIXIT                ;JUMP IF ZERO

MODRET: DJNZ R2,MODLP           ;UNTIL NUMBER OF INCREMENTS IS DONE

        MOV <CMODEL,A           ;STORE INCREMENTED MODEL

RETU:   RET                     ;RETURN FROM MODELC

FIXIT:  INC A                   ;INCREMENT FROM ZERO TO 1

        INC <ONE80              ;TOGGLE THE PHASE 180 BIT

        JMP MODRET              ;RETURN TO LOOP

ERROR:  JMP ERROR               ;ERROR TRAP FOR DIAGNOSTIC

ERROR3: MOV A,#3                ;ERROR CODE

        JMP ERROR               ;ERROR DIE

ERROR5: MOV A,#5                ;ERROR CODE

        JMP ERROR               ;ERROR DIE

ERROR7: MOV A,#7                ;ERROR CODE

        JMP ERROR               ;ERROR DIE

ERROR2: MOV A,#2                ;ERROR CODE

ERJ:    JMP ERROR               ;JUMP TO ERROR DIE

STORE:  MOV A,#A1LIML           ;GET VALUE

        MOV <A1LIM,A            ;STORE VALUE

        MOV A,#A1LIMH           ;GET VALUE

        MOV <A1LIM,A          ;STORE VALUE

        MOV A,#A2LIML           ;GET VALUE

        MOV <A2LIM,A            ;STORE VALUE

        MOV A,#A2LIMH           ;GET VALUE

        MOV <A2LIM,A          ;STORE VALUE

        MOV A,#SSC1L            ;GET VALUE

        MOV <SSC1,A             ;STORE VALUE

        MOV A,#SSC1H            ;GET VALUE

        MOV <SSC1,A           ;STORE VALUE

        MOV A,#SSC2L            ;GET VALUE

        MOV <SSC2,A             ;STORE VALUE

        MOV A,#SSC2H            ;GET VALUE

        MOV <SSC2,A           ;STORE VALUE

        MOV A,#SPHCL            ;GET VALUE

        MOV <SPHC,A             ;STORE VALUE

        MOV A,#SPHCH            ;GET VALUE

        MOV <SPHC,A           ;STORE VALUE

        RET                     ;RETURN FROM STORE

OUT:    MOV R5,A                ;TEMP

        MOV A,#CSOUT            ;GET CS

        CALL SETCS              ;GO SET CS

        MOV A,R5                ;GET SAVED VALUE

        MOV <IMAGE,A            ;SAVE IMAGE OF U33

        JNB ACC.1,RANGER         ;JUMP IF PHASE RANGE ERROR

        JNB ACC.2,RANGER         ;JUMP IF AMPLITUDE RANGE ERROR

        JNB ACC.3,RANGER         ;JUMP IF FREQUENCY RANGE ERROR

        ORL A,#00001000B        ;SET OUT OF RANGE LIGHT TO GREEN

OUT1:   MOVX @R0,A              ;LOAD OUT REGISTER

        MOV A,#CSOFF            ;CS ALL OFF

        CALL SETCS              ;GO SET CS

        RET                     ;RETURN FROM OUT

RANGER: ANL A,#11110111B        ;SET RANGE ERROR LIGHT TO RED

        JMP OUT1

SETCS:  ORL A,#11000000B        ;LEAVE COIL SELECT BITS AS ONES (INPUTS)

        MOV P1,A                ;SET NEW P1

        RET                     ;RETURN FROM SETCS

      ;

      ;SUBROUTINE TO INITIALIZE THE SCN2681 DUAL SERIAL COM CHIP

      ;

INTCOM: CALL CRESET             ;GO RESET COM CHIP

        MOV A,#CSCOM            ;GET CS

        CALL SETCS              ;SET COM CS

        MOV R0,#MR1A            ;ADDRESS OF MR1A

        MOV R2,#2               ;LOOP COUNTER

INLOOP: MOV A,#00010011B        ;MR1A/B SETUP VALUE

        MOVX @R0,A              ;LOAD MR1A/B

        MOV A,#00001111B        ;MR2A SETUP VALUE

        MOVX @R0,A              ;LOAD MR2A/B

        MOV R0,#MR1B            ;ADDRESS OF MR1B

        DJNZ R2,INLOOP          ;GO AROUND TWICE (A AND B)

        MOV R0,#CSRA            ;ADDRESS OF CSRA

        MOV A,#10111011B        ;9600 BAUD FOR RX AND TX

        MOVX @R0,A              ;LOAD CSRA REGISTER

        MOV R0,#CSRB            ;ADDRESS OF CSRB

        MOVX @R0,A              ;LOAD CSRB REGISTER

        MOV R0,#ACR             ;ADDRESS OF ACR

        CLR A                   ;ACR SET UP VALUE

        MOVX @R0,A              ;LOAD ACR

        MOV R0,#CRA             ;ADDRESS OF CRA

        MOV A,#01001001B        ;RESET ERRORS, EN RX

        MOVX @R0,A              ;LOAD CRA

        MOV R0,#CRB             ;ADDRESS OF CRB

        MOVX @R0,A              ;LOAD CRB

        MOV R0,#IMR             ;ADDRESS OF IMR

        MOV A,#00000010B        ;ENABLE INTERRUPTS 

        MOVX @R0,A              ;LOAD IMR

        MOV A,#CSOFF            ;CS OFF

        CALL SETCS              ;SET CS

        RET                     ;RETURN FROM INTCOM

CRESET: MOV A,<IMAGE            ;GET REGISTER IMAGE

        ORL A,#10000000B        ;SET COM RESET BIT

        CALL OUT                ;GO OUTPUT THE RESET BIT

        MOV A,<IMAGE            ;GET IMAGE AGAIN

        ANL A,#01111111B        ;RESET BIT TO ZERO

        CALL OUT                ;GO OUTPUT THE RESET BIT

        RET                     ;RETURN FROM CRESET

      ;INTERRUPT SERVICE SUBROUTINE

      ;SERVICES INTERRUPTS GENERATED BY COM CHIP

      ;REGISTERS (RB1):

      ;      R0 = GENERAL PURPOSE POINTER

      ;      R1 = GP INDIRECT POINTER

      ;      R2 = GP

      ;      R3 = RECEIVED CHARACTER

      ;      R4 = CHARACTER TO SEND

      ;      R5 = GP

      ;      R6 = 

      ;      R7 = 

      ;

INTSER: PUSH PSW                ;SAVE PSW ON STACK

        ORL PSW,#00001000B      ;SELECT BACKGROUND REGISTERS

        MOV R0,#<ISAVE5         ;POINT TO TEMP STORE

        MOV @R0,A               ;SAVE ACCUM.

        INC R0                  ;ADJUST POINTER

        MOV A,P1                ;GET P1 VALUE

        MOV @R0,A               ;SAVE P1

        INC R0                  ;ADVANCE POINTER

        MOV A,P2                ;GET P2 STATE

        MOV @R0,A               ;SAVE P2

        MOV A,#CSCOM            ;COM CHIP CS

        CALL SETCS              ;SET CS

        MOV R0,#SRA             ;ADDRESS OF SRA

        MOVX A,@R0              ;GET COM CHIP STATUS

        JB ACC.0,RECEIV         ;JUMP IF REC BUFFER FULL

        ;JB ACC.2,XMIT           ;JUMP IF TRANSMITTER READY FOR NEXT CHAR

        MOV A,#5                ;ERROR CODE

ERRORJ: JMP ERROR               ;IF NIETHER THERE IS AN ERROR

ERROR6: MOV A,#6                ;ERROR CODE

        JMP ERROR               ;JUMP TO ERROR DIE

RECEIV: JB ACC.4,ERROR6         ;JUMP IF OVERRUN ERROR

        MOV R0,#RHRA            ;ADDRESS OF RX DATA 

INT1:   MOVX A,@R0              ;GET REC CHAR FROM COM CHIP

        MOV R3,A                ;STORE CHAR

        MOV R0,#<STATUS         ;POINT TO STATUS

        MOV A,@R0               ;GET STATUS

        ORL A,#10000000B        ;SET CHAR REC'D BIT

        MOV @R0,A               ;STORE NEW STATUS

RETURN: MOV R0,#<ISAVE5       ;RETURNING NOW. POINT TO SAVE AREA 

        MOV A,@R0               ;GET SAVED P2

        MOV P2,A                ;RESTORE P2

        DEC R0                  ;NEXT SAVED 

        MOV A,@R0               ;GET SAVED P1

        MOV P1,A                ;RESTORE P1

        DEC R0                  ;NEXT

        MOV A,@R0               ;RESTORE ACCUM

        POP PSW                 ;RESTORE PSW

        RETI                    ;RETURN FROM INTERRUPT

        ;

MULTPY: ;MULTIPLIES A 4 BYTE NUMBER (MULIN4) BY A 12 BIT NUMBER (MULIN2) 

        ;AND GENERATES A 4 BYTE RESULT (ACCUM4)

        ;RETURNS CARRY = 1 FOR OVERFLOW ERROR

        CALL ZACC               ;ZERO OUT ACCUM4

        MOV R2,#12              ;SET MULT LOOP FOR 12 BIT MULTIPLY

MULOOP: CLR C                   ;CLEAR CARRY

        CALL SHIFT2             ;GO SHIFT MULIN2 RIGHT INTO CARRY

        JNC NOBIT               ;JUMP IF NO BIT IN THIS POSITION

        CALL ACCADD             ;GO ADD MULIN4 TO ACCUM4

        JC EXIT                 ;IF OVERFLOW IMMEDIATELY EXIT WITH C=1

        MOV A,R2                ;GET LOOP COUNTER

        XRL A,#1                ;TEST FOR LAST PASS

        JZ NOSHF                ;JUMP IF LAST PASS

NOBIT:  MOV R0,#<MULIN4         ;SET POINTER

        CALL SHIFT4             ;GO SHIFT MULIN4 LEFT

NOSHF:  DJNZ R2,MULOOP          ;UNTIL DONE WITH ALL BITS

EXIT:   RET                     ;RETURN FROM MULTPY

SHIFT2: MOV R0,#<MULIN2       ;SET POINTER

        MOV A,@R0               ;GET HI BYTE

        RRC A                   ;SHIFT RIGHT

        MOV @R0,A               ;STORE SHIFTED VALUE

        DEC R0                  ;NOW LO BYTE

        MOV A,@R0               ;GET LO BYTE

        RRC A                   ;SHIFT LO

        MOV @R0,A               ;STORE LO

        RET                     ;RETURN FROM SHIFT2

ACCADD: MOV R0,#<MULIN4         ;SET POINTER

        MOV R1,#<ACCUM4         ;SET POINTER

        MOV R3,#4               ;SET LOOP COUNTER

        CLR C                   ;CLEAR CARRY

ALOOP:  MOV A,@R0               ;GET BYTE

        ADDC A,@R1              ;ADD IN

        MOV @R1,A               ;STORE RESULT

        INC R0                  ;NEXT BYTE

        INC R1                  ;NEXT BYTE

        DJNZ R3,ALOOP           ;UNTIL ALL 4 BYTES ARE ADDED IN

        RET                     ;RETURN FROM ACCADD

SHIFT4: CLR C                   ;RESET CARRY

        MOV A,@R0               ;GET LO BYTE

        RLC A                   ;SHIFT RIGHT

        MOV @R0,A               ;STORE SHIFTED VALUE

        INC R0                  ;NOW MID LO BYTE

        MOV A,@R0               ;GET MID LO BYTE

        RLC A                   ;SHIFT MID LO

        MOV @R0,A               ;STORE MID LO

        INC R0                  ;NOW MID HI BYTE

        MOV A,@R0               ;GET MID HI BYTE

        RLC A                   ;SHIFT MID HI

        MOV @R0,A               ;STORE MID HI

        INC R0                  ;NOW HI BYTE

        MOV A,@R0               ;GET HI BYTE

        RLC A                   ;SHIFT HI

        MOV @R0,A               ;STORE HI

        RET                     ;RETURN FROM SHIFT4

ZACC:   ;HERE TO ZERO OUT ACCUM4

        CLR A                   ;GET A ZERO

        MOV R2,#4               ;SET LOOP COUNTER

        MOV R0,#<ACCUM4         ;POINT TO OUTPUT

ZLOOP:  MOV @R0,A               ;ZERO OUT OUTPUT

        INC R0                  ;NEXT

        DJNZ R2,ZLOOP           ;UNTIL DONE

        RET                     ;RETURN FROM ZACC

.END

